

## Paper 095-31

## ODS Statistical Graphics for Clinical Research

Wei Cheng, Isis Pharmaceuticals, Inc., Carlsbad, CA

### ABSTRACT

Graphical representations to illustrate the effects and safety of treatments are often more easily understood and remembered than tabular forms. Statistical graphics play an important role in clinical research. Beginning with SAS®9, a number of SAS/STAT procedures can use the Output Delivery System (ODS) to create statistical graphics automatically. You can use the ODS Graphics Template Language (GTL) to generate statistical graphics of your own layout design. This new facility which is referred to as ODS Statistical Graphics greatly enhances the ability to produce statistical graphs for clinical research. This paper describes how to create commonly used statistical graphs for clinical research using ODS Statistical Graphics:

- Create statistical graphs from SAS procedures;
- Customize statistical graphs using ODS templates and ODS styles;
- Create statistical graphs using ODS Graphics Template Language.

### INTRODUCTION

Graphs usually have greater impact and thus wider reception than tabular representations. Advanced technology of information visualization has resulted in the generation of more graphs with better quality. When we prepare the statistical summaries and analysis for clinical research, graphical displays help statisticians and clinicians more efficiently understand the data and the analysis. This paper presents how to apply the ODS Statistical Graphics (or ODS Graphics for short), which is a new data visualization facility available in SAS 9.1, to assist clinical research.

To understand many parts of this paper requires some statistical knowledge and clinical research background. This paper is for SAS programmers who are involved in clinical research, and it focuses on presenting analysis of clinical data graphically. The data sets for the examples in this paper are taken from Glenn A. Walker's book "Common Statistical Methods for Clinical Research with SAS® Examples" (Walker, 2002). Even though HTML file format is used as the ODS destination for the examples, other ODS destinations are supported by ODS Graphics such as RTF, PDF, etc.

ODS Statistical Graphics are experimental in SAS 9.1. It is not recommended for production work. In SAS 9.2, the ODS Graphics Template Language will become production software.

### CREATE STATISTICAL GRAPHS FROM SAS PROCEDURES

A number of SAS/STAT procedures such as BOXPLOT and LIFETEST can produce high-resolution graphical images since SAS Version 7. In SAS 9.1, in addition to these GRSEG-based graphs, over 15 SAS/STAT procedures can produce graphical displays that are template-based. The layout and appearance of ODS Graphics are controlled by ODS templates and styles. They support popular image formats GIF, JPEG, PNG, PS, EPSI, etc. ODS Graphics integrate the graphical displays fully with the tabular output. Graph production is automated by the procedure, and all ODS Graphics are governed by the standard ODS statements such as SELECT and EXCLUDE, just like ODS tables. ODS Graphics can generate different plot types with complex graphical layouts, and they are integrated with common file formats HTML, PDF, RTF, LATEX, etc.

### ODS GRAPHICS BASICS

ODS GRAPHICS statement Enables ODS automatic graphic capabilities. The basic syntax is

```
ODS GRAPHICS ON </ MAGEFMT = image-file-type | STATIC | STATICMAP
                IMAGENAME = filename RESET >;
                procedures or data steps
ODS GRAPHICS OFF;
```

Where

ODS GRAPHICS < OFF | ON > turns off / on the automatic ODS graphic generation;  
 IMAGEFMT specifies the image format to display graphics in ODS output, i.e. GIF, PNG, etc.;;  
 IMAGENAME specifies the base image filename;  
 RESET resets the index counter that is appended to static image files.

Common statistical methods for clinical research include the t-test, analysis of variance (ANOVA), regression, analysis of covariance (ANCOVA), the Chi-Square test, and the Log-Rank test, etc. Following examples are based on actual clinical trials (Walker, 2002).

#### EXAMPLE 1: REGRESSION ANALYSIS OF ANTI-ANGINAL RESPONSE VS. DISEASE HISTORY (WALKER, 2002 P178)

##### PROGRAM 1.1

```
data angina;
  input pat x_dur y_impr @@;
  datalines;
  1 1 40 2 1 90 3 3 30 4 2 30
```

```

.....
;
run;
ods html; ods graphics on / imagename = 'regression';
proc reg data = angina;
    model y_impr = x_dur;
run;quit;
ods graphics off; ods html close;

```

The REG procedure in the preceding program produces a scatter plot of the input data overlaid with the fitted regression line, confidence band, and prediction limits, as shown in Figure 1.1; a plot of the residuals versus the regressor in the model, as shown in Figure 1.2; and a summary panel, as shown in Figure 1.3, which consists of eight fit diagnostics plots:

- Residuals versus the predicted values
- Studentized residuals versus the predicted values
- Studentized residuals versus the leverage
- Normal quantile plot of the residuals
- Dependent variable values versus the predicted values
- Cook's D versus observation number
- Histogram of the residuals
- A "Residual-Fit" (or RF) plot consisting of side-by-side quantile plots of the centered fit and the residuals.

To display the plots individually instead of the default summary panel, specify the PLOTS(UNPACK) option in the PROC REG statement as follows:

#### PROGRAM 1.2

```

ods html; ods graphics on;
proc reg data = angina plots(unpack);
model y_impr = x_dur;
run; quit;
ods graphics off; ods html close;

```

FIGURE 1.1




FIGURE 1.2




FIGURE 1.3





FIGURE 1.4



You can use the ODS SELECT statement to select a particular subset of ODS graphs. The following statements will display the fit plot only, which is shown in Figure 1.1.

#### PROGRAM 1.3

```
ods html; ods graphics on; ods select fit;
proc reg data = angina;
  model y_impr = x_dur;
run;
quit;
ods graphics off; ods html close;
```

When you specify the HTML destination and the IMAGEFMT=STATICMAP option in the ODS GRAPHICS statement, the HTML output file is generated with tool tips.

#### PROGRAM 1.4

```
ods html; ods graphics on / imagefmt = staticmap;
ods select fit;
proc reg data = angina;
  model y_impr = x_dur;
run;
quit;
ods graphics off; ods html close;
```

Moving the mouse over a scatter plot displays a tool tip with detailed information for the corresponding patient, as shown in Figure 1.4.

You can use the GLM procedure to produce a grouped box plot of the response values versus the classification levels of the independent variable. The box plot can be generated by specifying a one-way analysis of variance model with just one independent classification variable. Figure 1.5 is generated by the following program:

#### PROGRAM 1.5

```
ods html; ods graphics on;
proc glm data = angina;
  class x_dur;
  model y_impr = x_dur;
run; quit;
ods graphics off; ods html close;
```

FIGURE 1.5





FIGURE 2.1



#### EXAMPLE 2: ANALYSIS OF COVARIANCE (ANCOVA) OF TRIGLYCERIDE CHANGES ADJUSTED FOR GLYCEMIC CONTROL (WALKER, 2002 P203)

Data are first analyzed using the one-way ANOVA, ignoring the covariate, by the following program. A box plot (Figure 2.1) will be generated, showing a non-significant treatment effect ( $p = 0.2858$  from SAS output).

#### PROGRAM 2.1

```
data tri;
  input trt $ pat hgbalc trichg @@;
  datalines;
FIB 2 7.0 5 FIB 4 6.0 10 FIB 7 7.1 -5
.....
;
```

```
run;

ods html; ods graphics on / imagename = 'anova';
proc glm data = tri;
class trt;
model trichg = trt / ss3;
run;
quit;
ods graphics off; ods html close;
```

PROC GLM is then used to conduct the ANCOVA, using the treatment (TRT) effect as a class variable and HGBA1C as a numeric covariate. An analysis of covariance plot is generated after submit the following program, as shown in Figure 2.2. The end result is a large difference in mean responses between supplements based on the adjusted values ( $p = 0.0048$  from SAS output).

#### PROGRAM 2.2

```
ods html; ods graphics on / imagename = 'ancova';
proc glm data = tri;
class trt;
model trichg = trt hgbalc / solution;
run; quit;
ods graphics off; ods html close;
```

FIGURE 2.2





FIGURE 3.1



### EXAMPLE 3: THE LOG-RANK TEST OF HSV-2 EPISODES FOLLOWING GD2 VACCINE (WALKER, 2002 P351)

The SAS/STAT procedure LIFETEST is used to perform the log-rank test in the following program:

#### PROGRAM 3.1

```
data hsv;
input vac $ pat wks x @@;
cens = (wks < 1);
wks = abs(wks);
trt = (vac = 'GD2');
datalines;
GD2 1 8 12 GD2 3 -12 10 GD2 6 -52 7
.....
;
run;

ods html; ods graphics on / imagename = 'lifetest';
proc lifetest data = hsv;
time wks * cens (1);
strata vac;
run;
ods graphics off; ods html close;
```


Figure 3.1 displays the graph of the product-limit survival function estimate versus the time (in weeks) to first recurrence of HSV-2 following immunization, as well as the summary statistics, generated from the preceding program.

If you want to view the other graphical displays of the survival distribution and related function estimates that PROC LIFETEST can produce, you can specify the PLOTS = option in the SURVIVAL statement:

PROGRAM 3.2

```
ods html; ods graphics on;
proc lifetest data = hsv;
  time wks * cens (1);
  strata vac;
  survival plots = (survival,density,epb,hazard,loglogs,logsurv,hwb,cl,stratum);
run;
ods graphics off; ods html close;
```

FIGURE 3.2 - FIGURE 3.9



Figures 3.2 to 3.9 are statistical graphics generated by the preceding program. They include a panel of plots for each stratum, a plot of the survival functions estimates for all strata, a plot of the Hall-Wellner bands for all strata, a plot of the equal precision bands for all strata, a plot of negative log of the estimated survival functions, and a plot of the log of the negative log of the estimated survival functions.

## CUSTOMIZE STATISTICAL GRAPHS USING ODS TEMPLATES AND ODS STYLES

The layout and details of graphs produced with ODS Graphics are controlled by graph template definitions (or graph definition for short). The graph definitions are written in the graph template language. The default graph templates supplied by SAS should suffice in most common applications of ODS Graphics. However, ODS Graphics enables you to modify the default template to customize the graphics when modifications are needed. The overall appearance of ODS Graphics is coordinated by style definitions, like ODS tables. Style definitions specify features such as fonts, sizes, color scheme, and line styles. You can create your own style definition or modify a style definition supplied by SAS to customize the appearance of ODS Graphics.

Example 4 demonstrates how to customize the layout of statistical graphs with templates. Example 5 demonstrates how to modify the appearance of the statistical graphs produced by SAS procedures.

### EXAMPLE 4: CUSTOMIZING LAYOUT OF ODS GRAPHICS

The REG procedure in Program 1.3 produces a scatter plot of the input data overlaid with the fitted regression line, confidence band, and prediction limits, as shown in Figure 1.1. This graph is controlled by the graph template Stat.Reg.Graphics.Fit. If you want to move the summary statistics (the box at the right) to the bottom of the graph, move the legend from the bottom to the top, and remove the title "Fit Plot for y\_imp", you can modify the template as Program 4.

For more examples of customizing the statistical graphs produced by SAS procedures, please refer to this author's SUGI 30 paper (Cheng, 2005).

#### PROGRAM 4

```
proc template;
  define statgraph Stat.Reg.Graphics.Fit;
    notes "Modified Fit Plot";
    dynamic _MODELLABEL _DEPLABEL _NObs _NParm _EDF _MSE _RSquare _AdjRSq _PREDLIMITS
      _CONFLIMITS _XVAR;
    layout lattice / rows=1 columns=1;
      ColumnHeader;
        layout gridded;
          DISCRETELEGEND "Fit" "Prediction" "Confidence" / across=3
            HALIGN=CENTER VALIGN=BOTTOM BORDER=ON background=GraphWalls:background;
        endlayout;
      EndColumnHeader;
    layout Overlay / yaxisopts=( label=_DEPLABEL );
      BAND ylimitupper=UPPERCL ylimitlower=LOWERCL x=_XVAR / fill=false
        lines=true sort=x fillcolor=StatGraphPredictionLimit:foreground
        name="Prediction" LegendLabel=_PREDLIMITS;
      BAND ylimitupper=UPPERCLMEAN ylimitlower=LOWERCLMEAN x=_XVAR /
        fill=true lines=false sort=x fillcolor=
          StatGraphConfidence:foreground name="Confidence" LegendLabel=_CONFLIMITS;
      SCATTERPLOT y=DEPVAR x=_XVAR / markersymbol=GraphDataDefault:markersymbol
        markercolor=GraphDataDefault:contrastcolor
        markersize=GraphDataDefault:markersize
        tip=( OBSERVATION _XVAR DEPVAR );
      SERIES y=PREDICTEDVALUE x=_XVAR / linecolor=StatGraphFitLine:contrastcolor
        linethickness=StatGraphFitLine:linethickness
        linepattern=StatGraphFitLine:linestyle
        sort=x name="Fit" LegendLabel="Fit";
    endlayout;
  Column2Header;
    layout gridded / columns=6 border=true background =GraphWalls:background;
      entry "NObs";entry "NParm";entry "EDF";
      entry "MSE";entry "RSquare";entry "AdjRSq";
      if (_NOBS=.) entry " NA ";
      else entry _NOBS / halign=center valign=bottom format=best3.;
      endif;
      if (_NPARM=.) entry " NA ";
      else entry _NPARM / halign=center valign=bottom format=best3.;
      endif;
      if (_EDF=.) entry " NA ";
      else entry _EDF / halign=center valign=bottom format=best3.;
      endif;
      if (_MSE=.) entry " NA ";
      else entry _MSE / halign=center valign=bottom format=best6.;
```


```

endif;
if (_RSQUARE=.) entry " NA ";
else entry _RSQUARE / halign=center valign=bottom format=best6.;
endif;
if (_ADJRSQ=.) entry " NA ";
else entry _ADJRSQ / halign=center valign=bottom format=best6.;
endif;
EndColumn2Header;
endlayout;
end;
run;

ods html; ods graphics on / imagename = 'modreg';
ods select fit;
proc reg data = angina;
model y_impr = x_dur;
run; quit;
ods graphics off; ods html close;

```

The result of the change, after submitting the modified template definition and the program in Program 4, is displayed in Figure 4:



#### EXAMPLE 5: CUSTOMIZING APPEARANCE OF ODS GRAPHICS

You can modify a style to customize the appearance of ODS Graphics in the same way as customizing the appearance of ODS tables. Four styles supplied by SAS are recommended for use with ODS Graphics: Analysis, Default, Journal, and Statistical. The style Default is used in this paper. Example 5 shows how to enlarge the fonts for the axis value (tick mark) labels, axis labels, titles, and footnotes if you want to emphasize them in the graph. The fit plot from REG procedure is used as an example.

#### PROGRAM 5

```

proc template;
define style bigfont;
parent = styles.default;
replace GraphFonts /
'GraphDataFont' = ("Arial",8pt) 'GraphFootnoteFont' = ("Arial",18pt,Bold)
'GraphTitleFont' = ("Arial",16pt,Bold) 'GraphValueFont' = ("Arial",12pt,Bold)
'GraphLabelFont' = ("Arial",16pt,Bold);
end;
run;

ods html style = bigfont; ods graphics on / imagename = 'modfont';
ods select fit;
proc reg data = angina;
model y_impr = x_dur;
run; quit;
ods graphics off; ods html close;

```

Submit the preceding program to display the fit plot with modified style, as shown in Figure 5.

## CREATE STATISTICAL GRAPHS USING ODS GRAPHICS TEMPLATE LANGUAGE

Starting in SAS 9.1, a new template type STATGRAPH, which describes the layout and appearance of a graph, was added to the ODS template type family. A stand alone graphics template can be defined, unassociated with any procedure. It's useful when a procedure does not produce a graph that you would like to see. You can define the graphics template, compile the template, use SAS procedures to create output data sets with analyzed results, and finally use a data step to feed the data to the graphics template via FILE and PUT statements.

### ODS GTL BASICS

In SAS 9.1, ODS GTL is a new language for defining graphs. The syntax is part of PROC TEMPLATE:

```
PROC TEMPLATE;
    DEFINE STATGRAPH name-of-graph-definition;
        <declaration-statements;>
        <layout-statements;>
        <plot-statements;>
        <text-statements;>
    END;
RUN;
```

The syntax of binding a graphics template to a SAS data set likes binding a table template to the DATA step to produce the table, ODS GRAPHICS statement is specified to request ODS Graphics:

```
ODS GRAPHICS ON;

DATA _NULL_;
    SET my-data;
    FILE PRINT
    ODS = (TEMPLATE = "my-graphics-template");
    PUT _ODS_;
RUN;
```

The following two examples demonstrate how to use ODS GTL to create statistical graphics templates and generate statistical graphics from DATA step.

### EXAMPLE 6: TWO-SAMPLE T-TEST OF FEV1 CHANGES (WALKER, 2002 P68)

In Program 6, a graphics template MEANCHG is defined by ODS GTL in PROC TEMPLATE. PROC SUMMARY is used to compute the mean change of FEV<sub>1</sub> and their confidence limits. Because computations are performed independent of the template, different statistics can be included or computed.

#### PROGRAM 6

```
proc format;
    value $trt 'A' = 'Active'
              'P' = 'Placebo';
run;
data fev;
    length trtgrp $ 7;
    input patno trt $ fev0 fev6 @@;
    chg = fev6 - fev0; if chg = . then delete; trtgrp = put(trt, $trt.);
    datalines;
    101 A 1.35 . 103 A 3.22 3.55 106 A 2.78 3.15
    .....
    ;
run;
proc template;
    define statgraph mygraphs.meanchg;
        layout gridded;
            entrytitle 'Bar Chart of Mean Change by Treatment Group' ;
            entrytitle 'with Upper and Lower CLM' ;
            barchartparm x = trtgrp y = chg_mean / yerrorupper=uclm yerrorlower=lclm;
        endlayout;
    end;
run;
ods html; ods graphics on / imagename='meanchange';
proc summary data = fev nway;
    class trtgrp;
    var chg;
```



```

output out = meanchg (drop = _type_ _freq_) mean = chg_mean lclm = lclm uclm = uclm;
run;
data _null_;
set meanchg;
label chg_mean = "Mean Change" trtgrp = "Treatment Group";
uclm = uclm - chg_mean; lclm = chg_mean - lclm;
file print ods = (template='mygraphs.meanchg');
put _ods_;
run;
ods graphics off; ods html close;

```

The two-sample t-test can be carried out by using PROC TTEST. The bar chart generated by the preceding program, as shown in Figure 6, can help to compare the mean changes visually.

FIGURE 6





FIGURE 7



#### EXAMPLE 7: MEAN WALKING DISTANCE OVER TIME (WALKER, 2002 P127)

Graph of mean measurement over time is one of the most commonly used statistical graphs to examine safety or efficacy of treatments. The following program uses the data from this study, computes summary statistics from PROC SUMMARY, defines a graphics template OVERTIME by ODS GTL, and generates a mean distance over time plot as shown in Figure 7.

#### PROGRAM 7

```

data wdvis;
input trt $ pat wd0 wd1 wd2 wd3 wd4;
datalines;
ACT 101 190 212 213 195 248
.....
;
run;
proc format;
value $trt 'ACT' = 'Active' 'PBO' = 'Placebo';
run;
data wdvisv;
length treatmnt $ 7;
set wdvis;
keep treatmnt pat month distance;
treatmnt = put(trt, $trt.); distance = wd0; month = 0; output;
distance = wd1; month = 1; output; distance = wd2; month = 2; output;
distance = wd3; month = 3; output; distance = wd3; month = 4; output;
run;
proc sort data = wdvisv; by treatmnt; run;
proc summary data = wdvisv nway; class month; var distance; by treatmnt;
output out = meandis (drop = _type_ _freq_) mean = dis_mean stderr = stderr;
run;
data meandis;
set meandis;
label month = "Month" dis_Mean = "Mean Distance";
elower = dis_mean - stderr; eupper = dis_mean + stderr;
run;
proc template;
define statgraph mygraphs.overtime;
layout gridded;

```

```

        entrytitle "Mean Distance Over Time" ;
        entrytitle '(Error Bars show ± SE)';
        layout overlay;
            scatterplot x = month y = dis_Mean
                / group = treatmnt name = "treat"
                yerrorlower = elower yerrorupper = eupper;
            seriesplot x = month y = dis_Mean / group = treatmnt;
        endlayout;
        discretelegend "treat";
    endlayout;
end;
run;
ods html; ods graphics on / imagename='overtime';
data _null_;
    set meandis;
    file print ods = (template='mygraphs.overtime');
    put _ods_;
run;
ods graphics off; ods html close;

```

## CONCLUSION

Graphical displays to illustrate the effects and safety of treatments in clinical research are very helpful to effectively present clinically relevant findings embedded in large volumes of data or complex study design. There are numerous types of statistical graphs which can assist statisticians and clinicians to interpret the results of a study easily and quickly. ODS Statistical Graphics is an exciting new facility provided by SAS 9. It greatly enhances the ability of graphical representations for clinical research. This paper only covers a few examples of statistical graphs commonly used in clinical research; the programs can be adapted to other statistical graphics with the same principle.

## REFERENCES

1. SAS Institute, Inc. (2002-2004), *SAS/STAT User's Guide, SAS OnlineDoc® 9.1.3*, Cary, NC: SAS Institute, Inc.
2. SAS Institute, Inc. (2005), "TEMPLATE Procedure: Creating ODS Statistical Graphics Output (Experimental)" <<http://support.sas.com/rnd/base/topics/statgraph/proctemplate/>> (Jan 21, 2005).
3. SAS Institute, Inc. (2005), "ODS Statistical Graphics (Experimental) and ODS Styles: Usage and Reference" <<http://support.sas.com/rnd/base/topics/statgraph/v91StatGraphStyles.htm>> (Jan 21, 2005).
4. Walker, Glenn A. (2002), "Common Statistical Methods for Clinical Research with SAS® Examples" Second Edition. Cary, NC: SAS Institute, Inc.
5. Cheng, Wei (2005), "SAS 8 Is Seen in the Rearview Mirror: Generate Statistical Graphics Using ODS in SAS® 9" *Proceedings of the 30th Annual SAS Users Group International (SUGI) Conference*.
6. Cartier, Jeff (2005), "The Power of the Graphics Template Language" *Proceedings of the 30th Annual SAS Users Group International (SUGI) Conference*.
7. Rodriguez, Robert N. (2004), "An Introduction to ODS for Statistical Graphics in SAS® 9.1" *Proceedings of the 29th Annual SAS Users Group International (SUGI) Conference*.

## CONTACT INFORMATION

I welcome and appreciate your comments and questions. Contact the author at:

Wei Cheng,  
 Isis Pharmaceuticals, Inc.,  
 1896 Rutherford Rd., Carlsbad, CA 92008  
 (760) 603-3807 Email: [wcheng@isisph.com](mailto:wcheng@isisph.com)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.